The Negative Binomial Probability distribution is an extension of the Geometric Distribution. In the Geometric Distribution one is asking about the probability of it taking x (independent) tries to get the first success.   In the Negative Binomial Distribution, one asks about the number of tries to get the r^th success.

If the x^th try produces the r^th success then there were r-1 successes distributed among x-1 tries. This is where the binomial comes in. There are "x-1 CHOOSE r-1" ways to do this. Furthermore, if p is the probability of "success" on an independent try. Then putting all of the pieces together yields Binomial[x-1,r-1]*p^r*(1-p)^(x-r).

Introduction to the Negative Binomial Distribution (jbstatistics)
https://www.youtube.com/watch?v=BPlmjp2ymxw

Geometric Distribution
Using Wolfram-Cloud/Mathematica to explore and simulate a Geometric Probability Distribution
https://www.youtube.com/watch?v=B2z2reF31oA

The Geometric Probability Distribution: Quartiles And Outliers using Wolfram-Cloud/Mathematica
https://www.youtube.com/watch?v=FVzby8SLhFw

An Introduction to the Geometric Distribution (jbstatistics)
https://www.youtube.com/watch?v=zq9Oz82iHf0

Start by testing that the probabilities sum to 1

```
In[1]:= Sum[Binomial[x - 1, r - 1] * p ^ r * (1 - p) ^ (x - r), {x, r, Infinity}]
Out[1]= 1
```

Multiply each probability by x to find the expected mean.

```
In[2]:= Sum[x * Binomial[x - 1, r - 1] * p ^ r * (1 - p) ^ (x - r), {x, r, Infinity}]
```

Out[2]= $\frac{r}{p}$

Let's do some of the steps ourselves to obtain the variance where we would multiply the probabilities by (x-r/p)^2
Take out one term, Express the Binomial in terms of factorials. Replace 1-p with y which we will set back to 1-p at the end.

In[3]:=
```
term = (x - r / p) ^ 2 * (x - 1)! / (x - r)! / (r - 1)! * p ^ r * y ^ (x - r)
```

Out[3]=
$$\frac{p^r \left(-\frac{r}{p}+x\right)^2 y^{-r+x}\,(-1+x)!}{(-1+r)!\,(-r+x)!}$$

In[4]:=
```
term2 = term /. x → n + r
```

Out[4]=
$$\frac{p^r \left(n+r-\frac{r}{p}\right)^2 y^n\,(-1+n+r)!}{n!\,(-1+r)!}$$

The sum of y^n/n! is the exponential.
We can replace (n+r-r/p)^2 with (y d/dy +r-r/p)^2
We can replace (n+r-1)! with the Gamma function integral of t^(n+r-1) Exp[-t] dt from 0 to Infinity

(y d/dy +r -r/p)^2 p^r /(r-1)! Integrate[t^(r-1) Exp[-t] t^n y^n/n!, {t,0, Infinity}]

The sum over n now gives Exp[ty]

In[5]:=
```
Integrate[t ^ (r - 1) * Exp[-t (1 - y)], {t, 0, Infinity}]
```

Out[5]=
$$(1-y)^{-r}\,\text{Gamma}[r] \quad \text{if} \quad \text{Re}[y] < 1\ \&\&\ \text{Re}[r] > 0$$

(y d/dy +r -r/p)^2 (p/(1-y))^r

In[9]:=
```
first[y_, r_, p_] := y D[(p / (1 - y)) ^ r, y] + (r - r / p) * (p / (1 - y)) ^ r
```

In[6]:=
```
bop = Simplify[ y D[(p / (1 - y)) ^ r, y] + (r - r / p) * (p / (1 - y)) ^ r]
```

Out[6]=
$$-\frac{r \left(\frac{p}{1-y}\right)^r (-1+p+y)}{p\,(-1+y)}$$

In[7]:=
```
bop2 = Simplify[y D[bop, y] + (r - r / p) * bop]
```

Out[7]=
$$\frac{r \left(\frac{p}{1-y}\right)^r \left(p^2\,y + r\,(-1+p+y)^2\right)}{p^2\,(-1+y)^2}$$

In[8]:=
```
bop3 = Simplify[y D[bop2, y] + (r - r / p) * bop2]
```

Out[8]=
$$-\frac{r \left(\frac{p}{1-y}\right)^r \left(p^3\,y\,(1+y) + 3\,p^2\,r\,y\,(-1+p+y) + r^2\,(-1+p+y)^3\right)}{p^3\,(-1+y)^3}$$

In[9]:=
```
bop4 = Simplify[y D[bop3, y] + (r - r / p) * bop3]
```

Out[9]=
$$\frac{r \left(\frac{p}{1-y}\right)^r \left(6\,p^2\,r^2\,y\,(-1+p+y)^2 + r^3\,(-1+p+y)^4 + p^4\,y\left(1+4\,y+y^2\right) + p^3\,r\,y\left(-4+4\,p+7\,p\,y+4\,y^2\right)\right)}{p^4\,(-1+y)^4}$$

In[10]:=
```
Simplify[bop2 /. y → 1 - p]
```
Out[10]=

$$\frac{r - p\, r}{p^2}$$

In[11]:=
```
cm2 = Sum[(x - r / p) ^ 2 * Binomial[x - 1, r - 1] * p ^ r * (1 - p) ^ (x - r), {x, r, Infinity}]
```
Out[11]=

$$\frac{r - p\, r}{p^2}$$

In[12]:=
```
Simplify[bop3 /. y → 1 - p]
```
Out[12]=

$$\frac{(-2 + p)\,(-1 + p)\, r}{p^3}$$

In[13]:=
```
cm3 = Sum[(x - r / p) ^ 3 * Binomial[x - 1, r - 1] * p ^ r * (1 - p) ^ (x - r), {x, r, Infinity}]
```
Out[13]=

$$\frac{2\, r - 3\, p\, r + p^2\, r}{p^3}$$

In[14]:=
```
Simplify[bop4 /. y → 1 - p]
```
Out[14]=

$$-\frac{(-1 + p)\, r\, \left(p^2 + 3\,(2 + r) - 3\, p\,(2 + r)\right)}{p^4}$$

In[15]:=
```
cm4 = Simplify[Sum[(x - r / p) ^ 4 * Binomial[x - 1, r - 1] * p ^ r * (1 - p) ^ (x - r), {x, r, Infinity}]]
```
Out[15]=

$$-\frac{1}{p^4}(-1 + p)\, r\, \left((-1 + p)\, r\, \left(8 + 6\, r + r^2 + p^2\, r^2 - 2\, p\, \left(2 + 3\, r + r^2\right)\right) + p^{4+r}\, \text{HypergeometricPFQ}[\{2, 2, 2, 1 + r\}, \{1, 1, 1\}, 1 - p]\right)$$

In[16]:=
```
sk = Simplify[bop3 / bop2 ^ (3 / 2) /. y → 1 - p]
```
Out[16]=

$$\frac{2 - p}{p\, \sqrt{\frac{r - p\, r}{p^2}}}$$

In[17]:=
```
ku = Simplify[bop4 / bop2 ^ 2 /. y →  1 - p]
```
Out[17]=

$$\frac{p^2 + 3\,(2 + r) - 3\, p\,(2 + r)}{r - p\, r}$$

In[18]:=
```
NumSamples = 10 000;
Samples = {};
r = 3;
p = 1/6;
For[i = 1, i ≤ NumSamples, i++,
StrikeNum = 0;
TryCount = 0;
While[StrikeNum < r,
rand = RandomVariate[UniformDistribution[{0, 1}]];
If[rand < p, StrikeNum++,];
TryCount++;
](* end while loop*)
AppendTo[Samples, TryCount];
](*end loop over samples*)
```

In[23]:=
```
N[Mean[Samples]]
```
Out[23]=
```
18.0835
```

In[24]:=
```
N[Variance[Samples]]
```
Out[24]=
```
90.7656
```

In[25]:=
```
N[cm2 /. {r → 3, p → 1/6}]
```
Out[25]=
```
90.
```

In[26]:=
```
N[Skewness[Samples]]
```
Out[26]=
```
1.14239
```

In[27]:=
```
N[sk /. {r → 3, p → 1/6}]
```
Out[27]=
```
1.1595
```

In[28]:=
```
N[Kurtosis[Samples]]
```
Out[28]=
```
4.78686
```

In[29]:=
```
N[ku /. {r → 3, p → 1/6}]
```
Out[29]=
```
5.01111
```

In[30]:=
```
Histogram[Samples, Automatic, "PDF"]
```

Out[30]=



In[31]:=
```
BoxWhiskerChart[Samples, "Outliers"]
```

Out[31]=



Attempt to get expression for the median -- not nice like the Geometric Distribution was

In[32]:=
```
Clear[p]; Clear[r];
```

In[33]:=
```
Solve[Sum[Binomial[x - 1, r - 1] * p ^ r * (1 - p) ^ (x - r), {x, r, M - r}] == 1 / 2, M]
```

Solve: This system cannot be solved with the methods available to Solve. Try Reduce or FindInstance instead.

Solve: This system cannot be solved with the methods available to Solve. Try Reduce or FindInstance instead.

Out[33]=

```
Solve[
 1 - (1 - p)^(M-2 r) p^r Binomial[-1 + M - r, -1 + r] (-1 + Hypergeometric2F1[1, M - r, 1 + M - 2 r, 1 - p]) == 1/2,
 M]
```